ФГБОУ ВПО «Воронежский государственный
технический университет»

Кафедра высшей математики
и физико-математического моделирования

# ОСНОВНЫЕ МАТЕМАТИЧЕСКИЕ ОПЕРАЦИИ В MAPLE

## МЕТОДИЧЕСКИЕ УКАЗАНИЯ

к выполнению лабораторных работ
по дисциплине «Информатика» по направлению 131000.62
«Нефтегазовое дело», профилю «Эксплуатация и обслуживание
объектов транспорта и хранения нефти, газа и продуктов
переработки», 150400.62 «Металлургия», профилю «Технология
литейных процессов», 140700.62 «Ядерная энергетика и тепло-
физика», профилю «Техника и физика низких температур»
очной формы обучения

Воронеж 2014

Составители: канд. техн. наук С.А. Кострюков,
канд. техн. наук В.В. Пешков,
канд. физ.-мат. наук Г.Е. Шунин

УДК 004.42+004.43

Основные математические операции в Maple: Методические указания к выполнению лабораторных работ по дисциплине «Информатика» по направлению 131000.62 «Нефтегазовое дело», профилю «Эксплуатация и обслуживание объектов транспорта и хранения нефти, газа и продуктов переработки», 150400.62 «Металлургия», профилю «Технология литейных процессов», 140700.62 «Ядерная энергетика и теплофизика», профилю «Техника и физика низких температур» очной формы обучения / ФГБОУ ВПО «Воронежский государственный технический университет»; сост. С.А. Кострюков, В.В. Пешков, Г.Е. Шунин. Воронеж, 2014. 17 с.

В методических указаниях кратко рассмотрены примеры выполнения простейших математических операций, необходимых студентам 1-го курса очной формы обучения, что позволяет использовать указания для всех форм обучения.

Издание соответствует требованиям Федерального государственного образовательного стандарта высшего профессионального образования по направлениям 131000.62 «Нефтегазовое дело», 150400.62 «Металлургия» и 140700.62 «Ядерная энергетика и теплофизика», профилям «Эксплуатация и обслуживание объектов транспорта и хранения нефти, газа и продуктов переработки», «Технология литейных процессов» и «Техника и физика низких температур» по дисциплинам «Информатика» и «Спецглавы информатики».

Методические указания подготовлены на магнитном носителе в текстовом редакторе Microsoft Word 2003 и содержатся в файле Maple-1курс.pdf.

Ил. 8. Библиогр.: 5 назв.

Рецензент канд. физ.-мат. наук, доц. В.В. Ломакин

Ответственный за выпуск зав. кафедрой д-р физ.-мат. наук, проф. И.Л. Батаронов

Издается по решению редакционно-издательского совета Воронежского государственного технического университета

© ФГБОУ ВПО «Воронежский государственный технический университет», 2014

# ВВЕДЕНИЕ

В настоящее время имеется значительное число программных продуктов, ориентированных на решение математических задач, как в символьной, так и в численной форме (MathCad, Maple, Mathematica, и т.д.). Пакет Maple является одним из лидеров среди универсальных систем и обеспечивает пользователю удобную и интеллектуальную среду для математических исследований.

Пакет Maple является мощной интегрированной системой, предназначенной для решения широкого круга математических задач, как в символьной, так и в численной форме. Он содержит более 2 тыс. встроенных функций по всем основным разделам математики, большинство из которых распределено по специализированным пакетам, загружаемым по требованию пользователя.

Пакет Maple создан совместно университетом Ватерлоо (штат Онтарио, Канада) и Высшей технической школой (ETH, Цюрих, Швейцария). Он широко распространен в университетах ведущих стран мира, исследовательских центрах и компаниях. При этом пакет Maple развивается, вбирая в себя новые умения, новые разделы математики и обеспечивая лучшую среду для работы.

Maple состоит из ядра – оптимизированных процедур, написанных на языке С, библиотек, написанных на Map1e-языке, и интерфейса. Ядро выполняет большинство базисных операций. Библиотеки содержат множество команд-процедур, выполняемых в режиме интерпретации. Пользователь имеет возможность программировать собственные процедуры, пополняя ими стандартный набор и, таким образом, расширяя возможности Maple.

В данных методических указаниях рассмотрены примеры выполнения простейших математических операций, необходимых студентам 1-го курса очной формы обучения.

# <u>1. Построение графиков</u>

В документе Maple команды вводятся в ячейке ввода после символа **>** (сам символ вводить не нужно). Команда завершается символом «**;**» (semicolon – точка с запятой) или «**:**» (colon – двоеточие). Двоеточие запрещает вывод на экран результатов команды (иногда это полезно).

Выполнение команд в ячейке ввода происходит только после нажатия клавиши Enter (это можно сделать в любой точке ячейки). Рекомендуется (особенно на первых этапах освоения программы) размещать в каждой ячейке ввода не более одной-двух команд.




Задаем функцию $f(x) = 0{,}5^x + 2 - (x-2)^2$:

```
> f:=x->0.5^x+2-(x-2)^2;
```

Строим график этой функции:

```
> plot(f(x),x=-20..20);
```

Строим новый график на более узком интервале, исправляя диапазон в предыдущей команде и нажимая Enter:




```
> plot(f(x),x=-10..20);
```

И еще раз:

```
> plot(f(x),x=-6.5..5);
```

Теперь видно, что график пересекает ось $x$ в трех точках, т.е. уравнение $f(x)=0$ имеет три корня.

Две кривые на одном графике, с указанием цветов:




```
> plot([3*sin(x),2^x],x=-2..2,
color=[red,blue]);
```

График функции, заданной в полярных координатах, с указанием толщины кривой («трехлепестковая роза»):

```
> plot([sin(3*x),x,x=0..2*Pi],
coords=polar,thickness=3);
```

График функции, заданной в параметрическом виде:

```
> plot([1-sin(t)^3,cos(t)^3,
t=0..2*Pi]);
```




Для построения графиков неявных функций нужно сначала подключить модуль **plots**, а затем использовать функцию **implicitplot**:

```
> with(plots):
> implicitplot({x^2+y^2=4,cos(x+y)=x*y},x=-3..3,y=-3..3);
```

Трехмерные графики:

```
> plot3d(x^2+y^2,x=-3..3,y=-3..3);
```

Для неявных функций:

```
> implicitplot3d(x^2+y^2-z^2=1,x=-1.2..1.2,
y=-1.2..1.2,z=-2..2, grid=[20,20,20]);
```








## Задания для самостоятельной работы

1. Построить графики функций:

$y = x^3 - 2x^2 + 1$; $y = \operatorname{arctg} x$; $y = \sqrt{x+5}\sin x$; $y = \ln|x|$; $y = \operatorname{th} x$.

2. Пересекаются ли графики функций:

а) $y = x^3 - 4x + 1$ и $y = \sqrt{x-2}$; б) $y = \ln(x^2+2)$ и $y = \sqrt[4]{x+5}$.

3. Построить графики функций, заданных параметрически:

а) $\begin{cases} x = t\cos t, \\ y = t\sin t, \end{cases}$ $t \in [0, 8\pi]$; б) $\begin{cases} x = \cos t + \sin 3t, \\ y = \sin t - \cos 3t, \end{cases}$ $t \in [0, 2\pi]$.

4. Построить совместно графики неявных функций:

$x^3 + xy^2 - y^3 + xy - y = 5$, $\ln(3 + 2x^2 + y) + xy = 2$.

5. Построить поверхности:

а) $x^2 + y^2 + z^2 = 1$; б) $z^2 = x^2 - y^2$; в) $z = 1 - e^{-x^2-y^2}$.

## 2. Решение уравнений

Уравнение представляет собой выражение вида

$$f(x)=0, \qquad (1)$$

в котором $f(x)$ – некоторая функция, а $x$ является неизвестной переменной. Решением (корнем) этого уравнения называется всякое число $x$ из области определения $f(x)$, которое, будучи подставлено в уравнение (1), обращает его в тождество. Геометрический смысл корня – это точка пересечения графика функции $f(x)$ с осью $Ox$.

Решить уравнение – значит найти совокупность (множество) всех его корней. Иногда задача ставится о решении уравнения лишь на каком-либо подмножестве числовой оси $x$ (например, на отрезке), тогда ищется совокупность корней уравнения, принадлежащих этому подмножеству.

Наибольшие сложности вызывает решение нелинейных уравнений. Их можно разделить на два класса – алгебраические и трансцендентные. *Алгебраическими* называются уравнения, содержащие только алгебраические функции (целые, рациональные, иррациональные). В частности, многочлен (полином) является целой алгебраической функцией. Уравнения, содержащие другие функции (тригонометрические, показательные, логарифмические и др.), называются *трансцендентными*.

Алгебраические уравнения решаются с помощью функции **solve**:

```
> solve(x^3-2*x+1,x);
```

$$1,-\frac{1}{2}+\frac{1}{2}\sqrt{5},-\frac{1}{2}-\frac{1}{2}\sqrt{5}$$

Как само уравнение, так и результат можно присвоить переменной:

```
> eq:= x^4-5*x^2+6*x=2:
> sols:= [solve(eq,x)];
```

$$sols := [-1+\sqrt{3}\,, -1-\sqrt{3}\,, 1, 1]$$

```
> sols[1];
```

$$-1+\sqrt{3}$$

```
> evalf(sols);
```

$$[.732050808, -2.732050808, 1., 1.]$$

С помощью функции **evalf** можно вывести результат в численном виде:

```
> solve(sqrt(ln(x))=2,x);
```

$$e^4$$

```
> evalf(");
```

$$54.59815003$$

Символ " (кавычка) в Maple означает последний вычисленный результат (по времени, а не по расположению команд).

Решение полиномиального уравнения:

```
> solve(x^5-3*x^4+2*x^2-x+3,x);
```

$$\mathrm{RootOf}(\_Z^5 - 3\,\_Z^4 + 2\,\_Z^2 - \_Z + 3)$$

```
> allvalues(");
```

$$-1.127479307, .03687403922 - .8565945569\,I,$$
$$.03687403922 + .8565945569\,I, 1.327862375, 2.725868853$$

Решить трансцендентное уравнение с помощью функции **solve**, как правило, не удается. Например, график функций $e^x$ и $\sin x$ показывает наличие бесконечного множества корней уравнения $e^x = \sin x$ при $x < 0$:

```
>plot([exp(x),sin(x)],x=-10..1);
```



Однако **solve** выдает комплексный корень:

```
> solve({exp(x)=sin(x)},x);
```

$$\{x = \mathrm{RootOf}(\_Z - \ln(\sin(\_Z)))\}$$

```
> allvalues(");
```

$$\{x = .3627020561 - 1.133745919\, I\}$$

Трансцендентные уравнения решаются приближенно с помощью функции **fsolve**:

```
> fsolve(exp(x)=sin(x),x=-4..0);
```

$$-3.183063012$$

Здесь нужно задавать отрезок, на котором расположен корень. Чтобы его найти, перед решением обычно строят график функции. Найдем еще один корень, теперь на отрезке [–7, –4]:

```
> fsolve(exp(x)=sin(x),x=-7..-4);
```

$$-6.281314366$$

Решение системы алгебраических уравнений:

```
> solve({x^2*y^2=0, x-y=1});
```

$$\{y = -1, x = 0\}, \{y = -1, x = 0\}, \{x = 1, y = 0\}, \{x = 1, y = 0\}$$

Для удобства левым частям уравнений системы можно присвоить имена:

```
> f:=x^7-5*x^2*y^4+1510:
> g:=y^5-3*x^4*y-105:
> s1:=solve({f,g},{x,y}); #Результат решения не приводится
> evalf(");
```

$$\{x = -2.844483289, y = -.5348543088\}$$

```
> allvalues(s1);  # Ниже приводится только часть результата
```

$$\{x = -2.844483289, y = -.5348543088\},$$
$$\{y = -2.573256586, x = -2.304767679\},$$
$$\{y = .1857181696+2.786617077\, I,\ x = -2.107398990-.1931448896\, I\}$$

- - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - -

$$\{x = 15.00039270, y = 19.74216374\}$$

Решение системы трансцендентных уравнений тоже обычно начинается с построения графика:

```
> with(plots):
```

```
> implicitplot({sin(x+y)-exp(x)*y=0,x^2-y=2},
x=-3..3, y=-3..3);
> fsolve({sin(x+y)-exp(x)*y=0,x^2-y=2},{x,y},
{x=-1..1,y=-2..0});
```

$$\{x = -.6687012050,\ y = -1.552838698\}$$

Для решения полиномиальных уравнений можно использовать функцию **roots**:

```
> roots(2*x^3+11*x^2+12*x-9);
```

$$[[-3, 2], [\frac{1}{2}, 1]]$$

В скобках первое число – корень, второе – его кратность.

## Задания для самостоятельной работы

1. Решить нелинейные уравнения, предварительно построив графики:

а) $5x^3 + 2x^2 + x - 1 = 0$; б) $x^6 - 3x^2 + 1 = 0$;

в) $\sqrt{x+1} - 1/x = 0$; г) $x^2 - \cos^2 \pi x = 0$;

д) $\mathrm{ch}(0{,}4x) = (x-2)^2$; е) $e^x - 6x - 3 + \mathrm{tg}\,x = 0$, $x \in [-\pi, \pi]$.

2. Решить системы уравнений, предварительно построив графики:

а) $\sin(x - y) - xy = -1, \quad x^2 - y^2 = 3/4$;

б) $x + 2\ln x - y^2 = 0, \quad 2x^2 - xy - 5x + 1$;

в) $\cos(x+5) - xy = 2.5, \quad \ln x + y^2 = 3$.

# 3. Задачи линейной алгебры

Большинство функций, предназначенных для решения задач линейной алгебры, находится в пакете **linalg**, который требуется подключить:

```
> with(linalg):
```

Задание квадратной матрицы 3-го порядка:

```
> A:=matrix([[2,1,3],[5,1,0],[7,8,9]]);
```

$$A := \begin{bmatrix} 2 & 1 & 3 \\ 5 & 1 & 0 \\ 7 & 8 & 9 \end{bmatrix}$$

Вычисление определителя матрицы:

```
> det(A);
```

$$72$$

Вычисление минора элемента $a_{21}$ матрицы:

```
> minor(A,2,1);
```

$$\begin{bmatrix} 1 & 3 \\ 8 & 9 \end{bmatrix}$$

Вычисление ранга матрицы:

```
> rank(A);
```

$$3$$

Вычисление следа матрицы (т.е. суммы диагональных элементов):

```
> trace(A);
```

$$12$$

Транспонирование матрицы:

```
> AT:=transpose(A);
```

$$AT := \begin{bmatrix} 2 & 5 & 7 \\ 1 & 1 & 8 \\ 3 & 0 & 9 \end{bmatrix}$$

Вычисление обратной матрицы:

```
> A1:=inverse(A);
```

$$A1 := \begin{bmatrix} \frac{1}{8} & \frac{5}{24} & \frac{-1}{24} \\ \frac{-5}{8} & \frac{-1}{24} & \frac{5}{24} \\ \frac{11}{24} & \frac{-1}{8} & \frac{-1}{24} \end{bmatrix}$$

Умножение матриц (используется специальный символ &*)

```
> evalm(A&*A1);
```

$$\begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

То же самое можно найти командой

```
> multiply(A,A1);
```

Одной из важнейших задач линейной алгебры является решение систем линейных алгебраических уравнений (СЛАУ).

Пусть требуется решить систему уравнений

$$\mathbf{Ax} = \mathbf{b},$$

где матрица **A** – введенная ранее, а вектор правых частей **b** равен $\begin{pmatrix} 5 \\ 6 \\ 1 \end{pmatrix}$. Сначала осуществим ввод вектора **b**:

```
> b:=matrix(3,1,[5,6,1]):
```

Вектор правых частей здесь можно задавать либо как матрицу размером 3×1, либо непосредственно как вектор (**vector**):

```
> b:=vector([5,6,1]):
```

*а) решение линейной системы с помощью функции* ***linsolve***

```
> linsolve(A,b);
```

$$\left[\frac{11}{6}, \frac{-19}{6}, \frac{3}{2}\right]$$

*б) решение линейной системы с помощью обратной матрицы*

```
> multiply(inverse(A),b);
```

*в) решение линейной системы методом Гаусса*

Вводим расширенную матрицу:

```
> Z:=matrix([[2,1,3,5],[5,1,0,6],[7,8,9,1]]);
```

$$Z := \begin{bmatrix} 2 & 1 & 3 & 5 \\ 5 & 1 & 0 & 6 \\ 7 & 8 & 9 & 1 \end{bmatrix}$$

```
> rref(Z);
```

В последнем столбце результата – решения системы.

Зададим векторы **u** и **v**:

```
> u:=vector([2,3,5]): v:=vector([-2.3,4,10]):
```

Скалярное произведение векторов:

```
> dotprod(u,v);
```

$$57.4$$

Векторное произведение:

```
> crossprod(u,v);
```

$$[10, -31.5, 14.9]$$

Угол между векторами в радианах:

```
> psi:=angle(u,v);evalf(psi);
```

$$\psi := \arccos(0.1371563117\sqrt{38})$$

$$0.5633174125$$

## Задания для самостоятельной работы

1. Вычислить:

а) $\begin{pmatrix} 2 & 1 & -2 \\ 0 & -1 & 4 \end{pmatrix} \cdot \begin{pmatrix} 3 & 1 & -2 \\ 0 & 3 & 4 \\ -1 & 2 & 2 \end{pmatrix}$; б) $\begin{pmatrix} 2 & 3 \\ -1 & 1 \end{pmatrix} \cdot \begin{pmatrix} 2 & 1 & 0 \\ -4 & 7 & 1 \end{pmatrix}$.

2. Решить СЛАУ: а) методом Гаусса; б) с помощью обратной матрицы; в) с помощью функции **`linsolve`**:

$$\begin{cases} 2x_1 - x_2 + 3x_3 = 1, \\ x_1 + 4x_2 - 2x_3 = -5, \\ 3x_1 + 2x_2 + x_3 = -2. \end{cases}$$

3. Для матрицы системы из задачи 2 найти определитель, транспонированную матрицу, след, ранг.

4. Для векторов $\vec{p} = \{2, -1, 4\}$, $\vec{q} = \{-3, 2, -2\}$, $\vec{r} = \{2, 1, -1\}$ найти смешанное и двойное векторное произведения.

## 4. Операции математического анализа

1) Дифференцирование

Вычисление производной функции $x^2 \operatorname{arctg}(x-2)$ может быть оформлено через функцию **diff**:

```
> diff(x^2*arctan(x-2),x);
```

$$2\,x\arctan(x-2)+\frac{x^2}{1+(x-2)^2}$$

Обратите внимание на следующую возможность вывода результата (слева так называемая инертная форма функции):

```
> Diff(x^2*arctan(x-2),x)=diff(x^2*arctan(x-2),x);
```

$$\frac{d}{dx}(x^2\arctan(x-2))=2\,x\arctan(x-2)+\frac{x^2}{1+(x-2)^2}$$

Производная восьмого порядка:

```
> Diff(x^2*ln(x-2),x$8)=diff(x^2*ln(x-2),x$8);
```

$$\frac{d^8}{dx^8}(x^2\ln(x-2))=-\frac{6720}{(x-2)^6}+\frac{11520\,x}{(x-2)^7}-\frac{5040\,x^2}{(x-2)^8}$$

Можно попытаться упростить этот результат:

```
> simplify(");
```

$$\frac{d^8}{dx^8}(x^2\ln(x-2))=-\frac{240\,(x^2-16\,x+112)}{(x-2)^8}$$

Частная производная по $x$ функции двух переменных $\cos(x/y+1)\,x^2y$:

```
> diff(cos(x/y+1)*x^2*y,x);
```

$$-\sin\left(\frac{x}{y}+1\right)x^2+2\cos\left(\frac{x}{y}+1\right)x\,y$$

Частная производная по $x$ 2-го порядка:

```
> Diff(cos(x/y+1)*x^2*y,x$2)=diff(cos(x/y+1)*x^2*y,x$2);
```

$$\frac{\partial^2}{\partial x^2}\cos\left(\frac{x}{y}+1\right)x^2\,y=-\frac{\cos\left(\frac{x}{y}+1\right)x^2}{y}-4\sin\left(\frac{x}{y}+1\right)x+2\cos\left(\frac{x}{y}+1\right)y$$

Для создания функций с производными может использоваться дифференциальный оператор **D**. В форме **D(f)(x)** этот оператор подобен **diff(f(x),x)**.

Поставим целью определить функцию **u(x)**, в каждой точке равную производной функции $f(x) = \sin x^2$. Применение **diff**, как и следует ожидать, дает нужное выражение

```
> fun:=x->sin(x^2);
> diff(fun(x),x);
```

$$2\cos(x^2)\,x$$

Но возникает вопрос: как это выражение присвоить имени функции, к которой потом можно было бы обращаться как к настоящей функции, т.е. вычислять значение при заданном аргументе, строить графики и т.д.? Такой, казалось бы, очевидный способ не приводит к успеху:

```
> u:=x->diff(fun(x),x);
```

$$u := x \to \mathrm{diff}(\mathrm{fun}(x), x)$$

```
> u(2.);
Error, (in u) wrong number (or type) of parameters in function diff
```

Проблема полностью решается с помощью оператора **D**:

```
> u:=D(fun);
```

$$u := x \to 2\cos(x^2)\,x$$

```
> u(1.); #
```

*значение функции* $f'(x)$ *в точке* $x=1$

$$1.080604612$$

2) Интегрирование

а) Неопределенные интегралы

Для вычисления неопределенных интегралов в Maple имеется функция **int**. При этом если аналитического значения интеграла не существует, возвращается исходная запись.

```
> int(a*x^n,x);
```

$$\frac{a\,x^{(n+1)}}{n+1}$$

Можно использовать инертную форму функции **int**:

```
> Int(ln(x)^3,x);
```

$$\int \ln(x)^3\,dx$$

В таком представлении интеграл записывается, но не вычисляется, а если его всё же требуется вычислить, то используется функция **value**:

```
> value(");
```

$$\ln(x)^3 x - 3x\ln(x)^2 + 6x\ln(x) - 6x$$

```
> Int(x^2*sin(x),x)=int(x^2*sin(x),x);
```

$$\int x^2 \sin(x)\,dx = -x^2\cos(x) + 2\cos(x) + 2x\sin(x)$$

`> Int(sin(x)/x,x)=int(sin(x)/x,x);` *# эта функция называется интегральный синус*

$$\int \frac{\sin(x)}{x}\,dx = \mathrm{Si}(x)$$

б) Определенные интегралы

Для вычисления определенных интегралов используются те же функции **int** и **Int**, в которых надо указать пределы интегрирования, например, **x=a..b**, если интегрируется функция переменной **x**.

```
> Int(sin(x)/x,x=a..b)=int(sin(x)/x,x=a..b);
```

$$\int_a^b \frac{\sin(x)}{x}\,dx = \mathrm{Si}(b) - \mathrm{Si}(a)$$

```
> int(sin(x)/x,x=0..1);
```

$$\mathrm{Si}(1)$$

Maple предпочитает выводить результат в таком как бы незавершенном виде, так как считает его более точным значением, чем в виде десятичного числа. Однако если требуется именно численное значение интеграла, причем с любой заданной точностью, можно применить функцию **evalf**. Существует и другая возможность сразу получить численный результат – записать один из пределов в виде вещественного числа, т.е. с точкой:

```
> int(sin(x)/x,x=0..1.);
```

$$0.9460830704$$

### 3) Вычисление пределов функций

Для вычисления предела функции в точке $x=a$ в пакете Maple используются функции

```
limit(f,x=a);                limit(f,x=a,dir);
Limit(f,x=a);                Limit(f,x=a,dir);
```

Здесь **f** – алгебраическое выражение, **x** – имя переменной, **dir** – параметр, указывающий направление поиска предела (**right** – справа, **left** – слева, **real** – в области вещественных значений, **complex** – в области комплексных значений). Значением **a** может быть бесконечность (**infinity**).

```
> Limit(sin(x)/x,x=0)=limit(sin(x)/x,x=0);
```

$$\lim_{x \to 0} \frac{\sin(x)}{x} = 1$$

```
> Limit(((x-2)/x)^x,x=infinity)=limit(((x-2)/x)^x,
x=infinity);
```

$$\lim_{x \to \infty} \left(\frac{x-2}{x}\right)^x = \mathbf{e}^{(-2)}$$

```
> Limit(2^(1/x),x=0,left)=limit(2^(1/x),x=0,left);
```

$$\lim_{x \to 0-} 2^{\left(\frac{1}{x}\right)} = 0$$

```
> Limit(2^(1/x),x=0,right)=limit(2^(1/x),x=0,right);
```

$$\lim_{x \to 0+} 2^{\left(\frac{1}{x}\right)} = \infty$$

```
> Limit((2^x-1)/x,x=-infinity)=limit((2^x-1)/x,
x=-infinity);
```

$$\lim_{x \to (-\infty)} \frac{2^x - 1}{x} = 0$$

```
> Limit((2^x-1)/x,x=infinity)=limit((2^x-1)/x,
x=infinity);
```

$$\lim_{x \to \infty} \frac{2^x - 1}{x} = \infty$$

4) Вычисление сумм и произведений

Для вычисления сумм и произведений последовательностей могут использоваться функции:

```
sum(f,k);                product(f,k);
sum(f,k=m..n);           product(f,k=m..n);
sum(f,k=alpha);          product(f,k=alpha);
```

Здесь **f** – функция, задающая члены суммируемого ряда, **k** – индекс суммирования, **m** и **n** – целочисленные пределы изменения **k**, **alpha** – выражение формата **RootOf**. Значение **n** может быть равно бесконечности (**infinity**).

```
> sum(k^2, k=0..4);
```

$$30$$

```
> Sum(k^2, k=0..4)=sum(k^2, k=0..4);
```

$$\sum_{k=0}^{4} k^2 = 30$$

Разработчики Maple рекомендуют при использовании упомянутых функций заключать **k** и **f** в прямые апострофы, например, **sum('f','k'=m..n)**. Это предотвратит возможную ошибку, связанную с предшествующим присваиванием переменной **k** определенного значения.

```
> sum('k^2', 'k'=0..n);
```

$$\frac{1}{3}(n+1)^3 - \frac{1}{2}(n+1)^2 + \frac{1}{6}n + \frac{1}{6}$$

```
> sum('a[k]*x^k','k'=0..4);
```

$$a_0 + a_1 x + a_2 x^2 + a_3 x^3 + a_4 x^4$$

```
> sum('1/k!', 'k'=0..infinity);
```

$$\mathbf{e}$$

### Задания для самостоятельной работы

1. Найти производные:

а) $y'$, если $y = \ln(\cos(x^2+1))$;   б) $y^{\mathrm{IV}}$, если $y = x\ln(1-3x)$;

в) $y'_x$ и $y''_x$ для параметрически заданной функции:

$x(t) = \operatorname{arctg} t$, $y(t) = \ln(1+t^2)$ .

2. Вычислить интегралы:

а) $\displaystyle\int \frac{\sqrt{x^2+4}}{x}dx$ ;  б) $\displaystyle\int\limits_{-\infty}^{\infty} \frac{dx}{(x^2+2)^2(x^2+10)^2}$ ;  в) $\displaystyle\int\limits_0^{2\pi} \frac{dt}{(\sqrt{7}+\cos t)^2}$ .

3. Вычислить пределы:

а) $\lim\limits_{x\to 0} x^2 e^{1/x^2}$ ; б) $\lim\limits_{x\to 1-0} \dfrac{\ln(1-x)+\operatorname{tg}(\pi x/2)}{\operatorname{ctg}\pi x}$ ; в) $\lim\limits_{x\to 1+0}\left(\dfrac{1}{\ln x} - \dfrac{x}{\ln x}\right)$.

4. Найти: а) $\displaystyle\sum_{n=1}^{\infty}\frac{1}{n^4}$ ; б) $\displaystyle\sum_{n=0}^{\infty}\frac{x^{n+3}}{(n+1)(n+2)}$ ; в) $\displaystyle\prod_{k=1}^{10}\left(1-\frac{1}{k!}\right)$.

## Вариант 1

1. Решить уравнения и систему (построить график, из него найти количество корней и начальные приближения):

а) $x^3 + 5x^2 - 5x = 9$;

б) $\dfrac{2x}{\sqrt{x+2}} - 3x^3 + 2x - 1 = 0$;

б) $e^x - 6x - 3 + \ln(x+3) = 0$;

в) $\begin{cases} \sqrt{x^2 + y^2} - ye^x = y \\ x^3 - y^2 = 3x + 1 \end{cases}$

2. Найти:

а) $y'$ и $y''$, если $y = x^2 \operatorname{arctg} \sqrt{x}$;

б) $\displaystyle\int \frac{\cos x dx}{\sin x + \cos x}$;

в) $\displaystyle\int_1^3 \frac{dx}{x^4 \cdot \sqrt{1+x^2}}$;

г) $\displaystyle\lim_{x \to 1} \frac{1-x}{\ln x}$.

3. Решить СЛАУ. Для матрицы $A$ системы найти обратную $A^{-1}$ и транспонированную $A^T$ матрицы, определитель матрицы. Найти произведение $A^{-1}A^T$.

$$\begin{cases} 3x_1 + 2x_2 + x_3 = 5 \\ 2x_1 - x_2 + x_3 = 6 \\ x_1 + 5x_2 = -3 \end{cases}$$

4. Даны векторы $\vec{p} = (2;\ -3;\ 1)$, $\vec{q} = (-3;\ 1;\ 3)$, $\vec{r} = (4;\ 2;\ -2)$. Найти их смешанное произведение и двойное векторное произведение.

## Вариант 2

1. Решить уравнения и систему (построить график, из него найти количество корней и начальные приближения):

а) $2x^3 - 10x^2 + 5 = 0$;

б) $\dfrac{\sqrt{x+2}}{3x^2+1} - x^3 - 1 = 0$;

в) $e^{-x} + x^2 - 9 - \ln(x+8) = 0$;

г) $\begin{cases} \ln(x^2 + y^2) - 3\sin x = 2xy \\ x^2 - y^3 = 3xy + 1 \end{cases}$

2. Найти

а) $y'$ и $y''$, если $y = \dfrac{\cos x}{\sqrt{4+x^2}}$;

б) $\displaystyle\int \frac{dx}{\cos x + 2\sin x + 3}$;

в) $\displaystyle\int_1^4 \frac{\sqrt{x} - 1}{\sqrt[3]{x} + 1} dx$;

г) $\displaystyle\lim_{x \to 0} \frac{\cos x - \cos 2x}{1 - \cos x}$.

3. Решить СЛАУ. Для матрицы $A$ системы найти обратную $A^{-1}$ и транспонированную $A^T$ матрицы, определитель матрицы. Найти произведение $A^{-1}A^T$.

$$\begin{cases} x_1 + x_2 - 2x_3 = 6 \\ 2x_1 + 3x_2 - 7x_3 = 16 \\ 5x_1 + 2x_2 + x_3 = 16 \end{cases}$$

4. Даны векторы $\vec{p} = (3;\ 2;\ -1)$, $\vec{q} = (1;\ -1;\ 2)$, $\vec{r} = (2;\ 1;\ -1)$. Найти их смешанное произведение и двойное векторное произведение.

## БИБЛИОГРАФИЧЕСКИЙ СПИСОК

1. Говорухин В.Н., Цибулин В.Г. Введение в Maple. Математический пакет для всех. - М: Мир, 1997. - 208 с.

2. Дьяконов В.П. Maple 7: Учебный курс. - СПб.: Питер, 2002. - 672 с.

# ОСНОВНЫЕ МАТЕМАТИЧЕСКИЕ ОПЕРАЦИИ В MAPLE

## МЕТОДИЧЕСКИЕ УКАЗАНИЯ

к выполнению лабораторных работ
по дисциплине «Информатика» по направлению 131000.62 «Нефтегазовое дело», профилю «Эксплуатация и обслуживание объектов транспорта и хранения нефти, газа и продуктов переработки», 150400.62 «Металлургия», профилю «Технология литейных процессов», 140700.62 «Ядерная энергетика и теплофизика», профилю «Техника и физика низких температур» очной формы обучения

Составители:
Кострюков Сергей Александрович
Пешков Вадим Вячеславович
Шунин Генадий Евгеньевич

В авторской редакции

Компьютерный набор В.В. Пешкова

Подписано к изданию 28.01.2014.

Уч.-изд. л. 1,1. “С” .

ФГБОУ ВПО «Воронежский государственный
технический университет»
394026 Воронеж, Московский просп., 14